
Multivariate normal data

SAS IML

Generating Multivariate Normal Data

The general idea comes from Factor Analysis.

Anger questionnaire:

Suppose there are N subjects each with 15 responses on some scale.  Theoretically, the 15 responses can be grouped into three "factors" which correspond to different types of anger.

How do you do this?  The basic idea of factor analysis is to form linear composites of the responses such that items within each combination, or factor, are correlated with each other but each linear combination (factor) is independent of the others.

Take the raw data matrix, and get a correlation matrix R:

The correlation matrix is the basic unit for factor analysis. 

Factor analysis:  go from correlation matrix to create independent scores.  Old technique described by Cholesky in 1915, involves square root or diagonal factoring.

Let R = correlation matrix

Cholesky decomposition is to find the matrix such that

CT*C = R

then use C as the next step in FA.

Reverse is to take independent scores and created correlated data, or generate multivariate normal data.

If standard MVN, the covariance matrix = correlation matrix.  

If we know R (set by experimenter), then find Cholesky factors of R (let C be the Cholesky factors matrix).

Generate independent standard normal deviates in a matrix Z where the z's are the columns of the matrix and there are n rows (for generating n subjects' data).

D = Z*C = matrix of raw data.

Assume we want n observations, 4 variables per observation, with correlations:

R will be a 4 x 4 symmetric matrix.

Find C, such that CT*C = R

C is also called the "square root" matrix

C is 4 x 4

Generate Z 

D = Z*C = 

(nx4) * (4x4) = (nx4) matrix of raw data 

which is standard MVN

Example:

R        1       2       3       4



correlation matrix

 1   1.000    .070    .250    .600

 2    .070   1.000    .070    .250

 3    .250    .070   1.000    .070

 4    .600    .250    .070   1.000

C         1       2       3       4


C'*C = R

 1   1.000    .0700    .2500    .6000

 2     .000    .9975    .0526    .2085

 3     .000    .0000    .9668   -.0941

 4     .000    .0000    .0000    .7666

Z  (5 subjects):  







   -1.0945   -1.8428    0.9454    1.3668

   -0.3159    0.9029    0.1494   -1.4340

    2.0795    0.1197   -0.2099    0.6988

    0.0574   -0.9739   -1.9344    0.5830

    0.8686   -1.8117   -0.3702   -0.7850

D=Z*C

  -1.0945   -1.9149     0.5434    -0.0821

   -0.3159    0.8786    0.1130   -1.1146

    2.0795    0.2649    0.3232    1.8281

    0.0574   -0.9675   -1.9071    0.4603

    0.8686   -1.7465   -0.2361   -0.4235

Two problems:

How do you find C?

1.  Healy, M.J.R.  (1968)   Algorithm AS6:  Triangular decomposition of a symmetric matrix, Applied Statistics, 17: 195-197.

I have an algorithm that does this.

How do you manipulate matrices in SAS?

Proc IML   (Interactive Matrix Language)

Powerful but complicated;  not necessarily needed for many simulations.  (use data step simulations where possible)

allows matrix manipulations and flexible programming.

all numbers can be considered "matrices", so you don't have to have a full-grown matrix to use PROC IML.

Lots of functions specific to IML, such as 

  x` or t(x), which take the transpose of matrix X.   

  inv(x) takes the inverse of X. 

 C = root(R) will produce C, such that R = CT*C.   

Warning:

Some familiar functions are different in IML, so need to be sure you're getting what you ask for.

 A*B in IML means multiply matrix A by matrix B.

A#B means multiply each element in A by the corresponding element in B.

First example:

proc iml;

  n = 5;

  r = {1 .07 .25 .6, .07 1 .07 .25, .25 .07 1 .07, .6 .25 .07 1};

  print r;

  c = root(r);

  print c;

  z=j(n,4,.);

    do i = 1 to n;


  do j = 1 to 4;


   z[i,j]=rannor(-99);


  end;


end;

  print z;   

  d = z*c;

  print d;

  x1 = d[,1];  x2=d[,2]; x3=d[,3]; x4=d[,4];

  print x1 x2 x3 x4;

  create mvn var{x1 x2 x3 x4};

  append;

 quit;

proc print data = mvn; 

run;

IML statements in example:

1.  proc iml;

starts the iml process

2.  r = {1 .07 .25 .6, .07 1 .07 .25, .25 .07 1 .07, .6 .25 .07 1};

this defines each element of matrix r.  { } define the matrix.

commas are used to define rows

3.  print r;


this simply prints matrix r

4.  c = root(r);

special IML function

root() is the function that performs the Cholesky decomposition and returns the matrix c

5.  z = j(n,4,.);

j is an IML matrix command 

This defines "z" as a matrix with n rows, 4 columns, and specifies that each element is missing "."

6.  do i = 1 to n;  do j = 1 to 4;

         z[i,j] = rannor(-99);

     end; end;

This generates the standard normal deviate for each element of matrix z.

z[i,j] is the value (element) in row i, col j.

7.  d = z*c

this is the matrix multiplication command.

8.  x1=d[,1];  

This defines x1 as a variable with the values in matrix d, col. 1.

Notice the comma:  If you wanted to define y as a variable with the values in matrix d, row 2, you would use:   



y = d[2,];

9.  create mvn var{x1 x2 x3 x4};

This creates a sas data set called mvn and specifies which variables are to be included in  that dataset.

10.  append

This reads the values into mvn

11.  quit;


Stops the IML process

Try this:

generate multivariate normal data with 3 variables.  Assume n = 200.  (x1, x2, and y)

Let the correlation between x1 and x2 = .2, the correlation between x1 and y = .7, and the correlation between x2 and y = .4.

Suppose we want to predict y from x1 and x2 jointly.

Generate (in IML) a regression equation to do this.

 Need to:

generate the data

define X as the matrix of predictors

define Y as a vector of outcomes

compute 

a vector of regression coefficients.

Check this by running your data through proc reg or proc glm.

� EMBED Equation.3  ���

